Order this document by AN1752/D REV 1

Motorola Semiconductor Application Note

AN1752

Data Structures for 8-Bit Microcontrollers

By Brad Bierschenk Consumer Systems Group Applications Engineering Austin, Texas

Introduction

A data structure describes how information is organized and stored in a computer system. Although data structures are usually presented in the context of large computers, the same principles can be applied to embedded 8-bit processors. The efficient use of appropriate data structures can improve both the dynamic (time-based) and static (storage-based) performance of microcontroller software.

This application note presents data structures which are useful in the development of microcontroller software. The applications presented here are by no means absolute. One can find an infinite variety of ways to apply these basic data structures in a microcontroller application.

Strings

A string is a sequence of elements accessed in sequential order. The string data structure usually refers to a sequence of characters. For example, a message which is to be output to a display is stored as a string of ASCII character bytes in memory.



Storing Strings

A string of elements must be identified by a starting and ending address. A starting address for a string can be defined using an absolute address label or by using a base address of a group of strings and identifying particular strings with an offset into the group.

There are several methods of terminating string information. One common way of terminating a string is by using a special character to mark the end of the string. One terminating character to use is the value \$04, an ASCII EOT (end-of-transmission) byte.

Figure 1 shows an example of string data.




Figure 1. String Data Structure

Another method of terminating a string is to identify its length. Its length can then be used as a counter value, eliminating the need for an extra byte of storage for the end of the string.

A string of ASCII characters can be terminated without using an extra byte of storage by using the sign bit (most significant bit) as an indicator of the last byte of the string. Because ASCII character data is only seven bits long, the last byte of a string can be indicated by a 1 in its most significant bit location. When using this method, the programmer must be careful to strip off the sign bit before using the ASCII character value.

Accessing Strings

An efficient way of accessing a string is with the indexed addressing mode and the INCX or DECX instruction. Listing 1. String Storage Example and Listing 2. String Access Example illustrate this string storage scheme and how to use it.

Listing 1. String Storage Example

```
* Absolute string addresses
* One way of specifying string data
Message1 FCB
                         'This is a string'
            FCB
            FCB
                         'This is another string'
Message2
             FCB
                         $04
* Indexed string addressing
* Another way of specifying string data
             EQU
Msqs
            EQU
                        *-Msgs
Message3
                         'This is a string'
             FCB
             FCB
                        $04
                         *-Msqs
Message4
            EQU
             FCB
                         'This is another string'
              FCB
```

Listing 2. String Access Example

LoadMsg	T.DX	#Meggage	.1	:Offcet	into X		
*						 	
* A gener:	ic method o	of displaying an	entire	string.			
* String of	display cod	le					

LoadMsg	LDX	#Message1	;Offset into X
Loop	LDA	Messages,X	;Load character
	CMP	#\$04	;Check for EOT
	BEQ	StringDone	;End of string
	JSR	ShowByte	;Show character
	INCX		;Point to next
	BRA	Loop	

---- String storage code

*		
Messages	EQU	*
Message1	EQU	*-Messages
	FCB	'This is a string'
	FCB	\$04

Application Note

String Applications

Practical applications of strings include storing predefined "canned" messages. This is useful for applications which require output to text displays, giving users information or prompting users for input.

Strings are also effective for storing initialization strings for hardware such as modems. Strings may also store predefined command and data sequences to communicate with other devices.

Stacks

A stack is a series of data elements which can be accessed only at one end. An analogy for this data structure is a stack of dinner plates; the first plate placed on the stack is the last plate taken from the stack. For this reason, the stack is considered a LIFO (last in, first out) structure. The stack is useful when the latest data is desired. A stack will typically have a predefined maximum size.

shows a representation of a stack.




Figure 2. Stack Data Structure

Application Note Stacks

Just like a physical stack of items, the software stack has a bottom and a top. Software should keep track of the location of the top of the stack. This address can either point to the first piece of valid data or it can point to the next available location. For the following examples it will be pointing to the next available location.

Stack Reading and Writing

The read operation of a stack is called "pulling" (or "popping"), and the write operation of a stack is called "pushing." When one pulls data from the stack, the data is removed from the stack and the stack pointer is adjusted. When data is pushed onto the stack, the stack pointer is adjusted and the data is added to the stack.

So, in the implementation of **Figure 2**, a push operation would first decrement the stack pointer and then store the data to the address pointed to by (stack pointer) +1. A pull operation would retrieve the data at (stack pointer) +1 and then increment the stack pointer.

Two error conditions are intrinsic to this data structure; underflow and overflow. A stack underflow occurs when a user attempts to pull information off an empty stack. A stack overflow occurs when a user attempts to push information onto a stack which is full. When using this data structure, these conditions should be attended to. An underflow condition should return an error. On an overflow, one can either reject the data and return an error, or the stack can "wrap" around to the bottom, destroying the data at the bottom of the stack.

MCU Hardware Stack

Motorola MCUs utilize a stack structure for saving program context before transferring program control. This interaction may be the result of a jump or interrupt. As a result of an interrupt, the stack is used to push the values in the X, A, and CCR (condition code register) registers, as well as the 16-bit PC (program counter) value. When a jump instruction is encountered, the PC value is pushed on to the stack. On returning from an interrupt (RTI instruction) the program context (registers and PC) are pulled from the stack. When returning from a jump (RTS instruction) the PC is pulled from the stack.

Application Note

HC05 Stack

The HC05 Family of MCUs have limited stack access. The only operation that can be performed with the MCU's stack pointer is to reset it. The RSP instruction will reset the stack pointer to \$FF. The HC05 stack pointer also has a limited size of 64 bytes. When the stack pointer grows beyond address \$C0, the stack pointer wraps around to \$FF, destroying any existing data at that address.

HC08 Stack

The HC08 Family of MCUs has a more flexible stack structure. The stack pointer can be set to any address. The HC08 MCUs also have an added addressing mode which is indexed by the stack pointer. In this way, a user can pass parameters to subroutines using the hardware stack, accessing the parameters using stack pointer indexed addressing.

Other HC08 Family instructions allow data to pushed on and pulled off the stack. The stack pointer can also be transferred to the X index register and vice-versa. With the addition of these instructions and addressing modes, a user has good control over the stack in the HC08 MCU.

Stack Applications

A stack is useful for dynamically allocating memory or passing parameters to and from subroutines. Typically, MCU RAM variables are statically allocated at assembly time.

For example:

```
; Statically allocated RAM variables
                    RAMSPACE
           ORG
MyVar1
           RMB
                    1
MyVar2
                    1
           RMB
MyVar3
           RMB
           ; Another method to statically allocate variable
MyVar4
           EQU
                    RAMSPACE+4
MyVar5
           EQU
                    RAMSPACE+5
```

This is appropriate for global variables, which need to be available throughout the program flow. However, for local variables which are only used in specific subroutines, this method is not the most efficient. The RAM space these variables use can be dynamically allocated using a software stack or MCU stack, freeing up RAM memory. The same

Freescale Semiconductor, Inc.

Application Note Stacks

method can be applied to subroutine input and output parameters, passing them on the stack instead of in the A or X register.

Listing 3. Software Stack shows a software implementation of a stack, which would be appropriate for the HC05 Family of microcontrollers.

Application Note

Listing 3. Software Stack

* STACK.ASM					
			mply shows the PUSH and PULL operations or		
		nded to be a complete ap			
		o next (empty) available	e location		
		C68HC705P6A MCU			
* Memory map					
RAMSPACE	EQU	\$50			
ROMSPACE	EQU	\$100			
RESETVEC	EQU	\$1FFE			
*					
* Stack equa *					
" STACKSIZE		\$08			
STACKBOT			;Bottom of software stack		
			;Maximum address of stack		
*					
* RAM variak *					
^	ORG		;First address of RAM		
StackPtr	RMB		;Pointer to next stack byte		
*					
* * Start of p					
* Start of p	orogram (code 			
* Start of p *	orogram o	code ROMSPACE	;Start of ROM		
* Start of p *	orogram o ORG LDA	code ROMSPACE #STACKBOT	;Start of ROM		
* Start of p *	orogram o ORG LDA	code ROMSPACE	;Start of ROM		
* Start of r * Init *	orogram (ORG LDA STA	code ROMSPACE #STACKBOT StackPtr	;Start of ROM		
* Start of r * Init * * Some simpl	orogram (ORG LDA STA	code ROMSPACE #STACKBOT StackPtr and write operations	;Start of ROM ;Initialize the stack pointer		
* Start of r *	ORG LDA STA Le read a	ROMSPACE #STACKBOT StackPtr and write operations only	;Start of ROM ;Initialize the stack pointer		
* Start of r *	Orogram (ORG LDA STA Le read a	ROMSPACE #STACKBOT StackPtr and write operations only #\$01	;Start of ROM ;Initialize the stack pointer		
* Start of r *	ORG LDA STA e read a cration of LDA JSR	ROMSPACE #STACKBOT StackPtr and write operations only #\$01 PushA	;Start of ROM ;Initialize the stack pointer		
* Start of r *	ORG LDA STA e read a cration of LDA JSR BCS	ROMSPACE #STACKBOT StackPtr and write operations only #\$01 PushA FullErr	;Start of ROM ;Initialize the stack pointer ;Write to stack		
* Start of r *	ORG LDA STA e read a ration of LDA JSR BCS JSR	ROMSPACE #STACKBOT StackPtr and write operations only #\$01 PushA FullErr PushA	;Start of ROM ;Initialize the stack pointer		
* Start of r *	ORG LDA STA Le read a cration of LDA JSR BCS JSR BCS	ROMSPACE #STACKBOT StackPtr and write operations only #\$01 PushA FullErr PushA FullErr	;Start of ROM ;Initialize the stack pointer ;Write to stack ;Write to stack		
* Start of r *	ORG LDA STA Le read a cration of LDA JSR BCS JSR BCS JSR BCS JSR	ROMSPACE #STACKBOT StackPtr and write operations only #\$01 PushA FullErr PushA FullErr PushA FullErr PushA FullErr	;Start of ROM ;Initialize the stack pointer ;Write to stack		
* Start of r *	ORG LDA STA Le read a cration of LDA JSR BCS JSR BCS JSR BCS JSR BCS	ROMSPACE #STACKBOT StackPtr and write operations only #\$01 PushA FullErr PushA FullErr PushA FullErr PushA FullErr PushA FullErr	;Start of ROM ;Initialize the stack pointer ;Write to stack ;Write to stack ;Write to stack		
* Start of r *	ORG LDA STA LDA LOA LOA LOA LOA LOA LOA LOA LOA LOA LO	ROMSPACE #STACKBOT StackPtr and write operations only #\$01 PushA FullErr	;Start of ROM ;Initialize the stack pointer ;Write to stack ;Write to stack		
* Start of r *	ORG LDA STA LDA JSR BCS	ROMSPACE #STACKBOT StackPtr and write operations only #\$01 PushA FullErr PushA FullErr PushA FullErr PushA FullErr PushA FullErr PushA FullErr	;Start of ROM ;Initialize the stack pointer ;Write to stack ;Write to stack ;Write to stack ;Write to stack		
* Start of r * Init * * Some simpl * For illust	ORG LDA STA LDA JSR BCS JSR BCS JSR BCS JSR BCS JSR BCS JSR BCS JSR	ROMSPACE #STACKBOT StackPtr and write operations only #\$01 PushA FullErr	;Start of ROM ;Initialize the stack pointer ;Write to stack ;Write to stack ;Write to stack		
* Start of r * Init * * Some simpl * For illust	ORG LDA STA Le read a cration of LDA JSR BCS	ROMSPACE #STACKBOT StackPtr and write operations only #\$01 PushA FullErr PushA FullErr PushA FullErr PushA FullErr PushA FullErr PushA FullErr	;Start of ROM ;Initialize the stack pointer ;Write to stack ;Write to stack ;Write to stack ;Write to stack		
* Start of r * Init * * Some simpl * For illust	ORG LDA STA LE read a cration of LDA JSR BCS JSR	ROMSPACE #STACKBOT StackPtr and write operations only #\$01 PushA FullErr	<pre>;Start of ROM ;Initialize the stack pointer ;Write to stack ;Write to stack</pre>		
* Start of r * Init * * Some simpl * For illust	ORG LDA STA LE read a cration of LDA JSR BCS	ROMSPACE #STACKBOT StackPtr and write operations only #\$01 PushA FullErr	<pre>;Start of ROM ;Initialize the stack pointer ;Write to stack ;Write to stack</pre>		
* Start of r * Init *	ORG LDA STA LE read a cration of LDA JSR BCS JSR	ROMSPACE #STACKBOT StackPtr and write operations only #\$01 PushA FullErr	<pre>;Start of ROM ;Initialize the stack pointer ;Write to stack ;Write to stack</pre>		

	Daa	De 11 Dece	
	BCS JSR	FullErr PushA	;Write to FULL stack
	BCS	FullErr	/Wiite to roll stack
	JSR	PushA	;Write to FULL stack
	BCS	FullErr	
	JSR	PullA	Read from stack
	BCS	EmptyErr	
	JSR	PullA	;Read from stack
	BCS	EmptyErr	
	JSR	PullA	;Read from stack
	BCS	EmptyErr	
Loop	BRA	*	;Your code here
EmptyErr	BRA	*	;Your code here
FullErr	BRA	*	¡Your code here
* Subroutine	es - The d	code to access the	data structure
* PUSH subr			
		of the accumulator	onto stack
		o indicate full er:	
*			·
PushA	LDX	StackPtr	Get stack pointer
		#STACKMAX	Check for full stack
	BLO	Full	
	DECX		;Decrement stack pointer
	STA	1,X	;Store data
	STX	StackPtr	Record new stack pointer;
	CLC		Clear carry bit;
	RTS		;Return
Full	SEC		;Set carry bit for error
	RTS		;Return
*			
* PULL subr			
		e stack into accum	
* Use C bit	of CCR to	o indicate empty s	tack error
^		C+ 2 c1-D+ -2	·Cot stack mainten
PullA	LDX	StackPtr #grackPor	Get stack pointer; Check for empty stack
	CPX BEQ	#STACKBOT Empty	refleck for empty stack
	LDA	1,X	;Get data
	INCX	Δ,Δ	;Increment stack pointer
	STX	StackPtr	Record stack pointer
	CLC	DCGCKI CI	Clear carry bit
	RTS		Return
Empty	SEC		;Set carry bit
	RTS		Return
*			
* Vector de *	Ilnitions 		
	ORG	RESETVEC	
	FDB	Init	
AN1752 — RE'	V 1		
	• •		

Using the software stack, a subroutine can allocate variables by pushing (allocating) bytes on the stack, accessing them with indexed addressing (relative to the stack pointer variable) and pulling them (deallocating) before returning. In this way, the same RAM space can be used by multiple subroutines.

Parameters can be passed to and from subroutines as well. An input parameter can be pushed on the stack. When a subroutine is entered, it can access the input parameter relative to the stack pointer. By the same token, a subroutine can push an output parameter onto the stack to be passed back to the calling routine.

The MCU hardware stack and stack pointer can also be used for these purposes. Because of the expanded instruction set, the use of the MCU stack is easily exploited in the HC08 Family of microcontrollers.

Listing 4. Using the HC08 Stack Operations shows an example of using the HC08 MCU stack to pass parameters and allocate local variables.

Listing 4. Using the HC08 Stack Operations

Using the stack to pass parameters and allocate variables optimizes memory usage.

```
*-----
```

- * Code segment example of using the HC08 stack to pass parameters and
- * allocate local variables.
- * Not intended to be a complete application.

× – –	 	 	 	 	 	

	LDA PSHA PSHA	#\$AA	;Load some data to be passed ;Push parameter for subroutine ;Push parameter for subroutine
	JSR	Sub	;Call subroutine
	PULA		;Parameter passed back
	STA	Result2	
	PULA		;Parameter passed back
	STA	Result1	
Loop	BRA *		;Your code here

* * Subr	outine w	hich uses	the	 stack	for	variable	e access
*							
*							
* SP	->Empty						
*	LOCAL2						
*							
*	LOCAL1						
*	PCH						
*							
*	PCL						
*							
*	PARAM2						
*	PARAM1						
*							
PARAM1		EQU		6		; E	Parameters passed in
PARAM2		EQU		5			
LOCAL1		EQU		2		; I	local variables
LOCAL2		EQU		1			
Sub		PSHA				;	Allocate local variable
		PSHA				;	allocate local variable
		LDA		PARAM1	,SP	; I	load the parameter passed in
		ROLA				; [Oo something to it
		STA		LOCAL1	,SP		Store in a local variable
		LDA		PARAM2	,SP	; I	load the parameter passed in
		ROLA					
		STA		LOCAL2		; 5	Store in a local variable
		LDA		LOCAL1	-		
		STA		PARAM1	-	; 5	Store value to be passed back
		LDA		LOCAL2	•		
		STA		PARAM2	,SP		Store value to be passed back
		PULA					Deallocate local variable memory
		PULA					Deallocate local variable memory
		RTS				; F	Return

Queues

A queue is a series of elements which accepts data from one end and extracts data from the other end. An analogy for this data structure would be a checkout line at the supermarket. The first people in are the first people out. For this reason, it is considered a FIFO (first in, first out) structure. This is useful when accessing data in the order it is received. A queue will usually have a predefined maximum size.

Figure 3 illustrates a queue.




Figure 3. Queue

Reading and Writing

The read operation of a queue is called "dequeue," and the write operation is "enqueue." Two pointers are necessary for a queue, one for the head of the line, and one for the tail. For an enqueue operation, after checking the size of the queue, the data is stored at the location pointed to by the "put" pointer, and the put pointer is adjusted. For a dequeue operation, the data is read from the "get" pointer location, and the pointer is adjusted.

Queues usually have a fixed size, so it is important to keep track of the number of items in the queue. This can be done with a variable containing the size of the queue or with pointer arithmetic.

Application Note Queues

Queue Errors

As with the stack structure, a queue can be subject to underflow and overflow errors. The write, or "enqueue" operation, should be non-destructive and should error if the queue is full. The read, or "dequeue" operation, should be destructive (remove the data element) and should error if the queue is empty.

Queue Applications

A practical application of a FIFO queue is for a data buffer. Queues can be used as buffers for transmitted or received data and for use with printers or serial communication devices.

Listing 5. Queue Example shows an example of queue software. A good application for this would be to store data received from the SIOP (serial input/output port) for processing later.

Listing 5. Queue Example

- *_____
- * Illustrates an implementation of a queue For the 705P6A
- *_____
- * Register definitions
- * Memory map definitions

RAMSPACE EQU \$50 ROMSPACE EQU \$100 RESETVEC EQU \$1FFE

- * Queue data structure definitions
- * These three equates defines the data structure
- * To change the queue, change the data structure,
- * and not the code.

QMAX	EQU	! 4	;Maximum Q size
QTOP	EQU	\$A0	;Top of Q array
QBOT	EQU	QTOP+QMAX-1	;Bottom of Q array

* RAM variables

	ORG	RAMSPACE	
TempA	RMB	1	
TempX	RMB	1	
GetPtr	RMB	1	;8-bit pointer
PutPtr	RMB		;8-bit pointer
QCount	RMB	1	;Counter for Q size

Freescale Semiconductor, Inc.

Application Note

	program code		
	ORG	ROMSPACE	
Start	EQU	*	
InitQ	LDA	#QTOP	;Initialize Q pointers and variables
~	STA	GetPtr	~ -
	STA	PutPtr	
	CLR	QCount	
Write and A good a	d read from topplication of		
t	JSR	DeQ	
	LDA	#\$FF	
	JSR	#9FF EnQ	
	SR	EnQ EnQ	
	JSR	EnQ	
	JSR	EnQ	
	JSR	EnQ	
	JSR	DeQ	
	SR	DeQ	
	LDA	#\$55	
	JSR	EnQ	
	JSR	EnQ	
oop	BRA	*	
Subroutin	nes 		
		a byte passed in ac	ccumulator A carry bit if full.
Otherwise	e returns a d	cleared carry bit o	on successful enqueue.
lnQ	STX	TempX	Save X register contents
	LDX	QCount	;Check for a full Q
	OMDV	#QMAX	
	CMPX		. 0 . 5 . 3 . 3
	BEQ	QFull	;Q full error
		QFull PutPtr	Q full error
	BEQ		;Store the data in A
	BEQ LDX	PutPtr	
	BEQ LDX STA	PutPtr 0,X	;Store the data in A
	BEQ LDX STA CMPX	PutPtr 0,X #QBOT	;Store the data in A ;Check for wrap
	BEQ LDX STA CMPX BEQ	PutPtr 0,X #QBOT	;Store the data in A ;Check for wrap ;Wrap the put pointer
√rapPut	BEQ LDX STA CMPX BEQ INCX	PutPtr 0,X #QBOT WrapPut	;Store the data in A ;Check for wrap ;Wrap the put pointer

EnQDone ;Unsuccessfu	STX LDX INC CLC RTS	PutPtr TempX QCount	<pre>;Store new put pointer ;Restore X register ;Increment count variable ;Clear carry bit ;Return</pre>
QFull	LDX SEC RTS	TempX	<pre>;Restore X register ;Set carry bit ;Return</pre>
* If the que * return a s * return a c	ue is empty, et carry bit	to indicate an erro	
DeQ	STX LDX CMPX BEQ LDX LDA	TempX QCount #\$00 QEmpty GetPtr 0,X	;Save X register contents ;Check for empty Q
	CMPX BEQ INCX BRA	#QBOT WrapGet DeQDone	;Check for wrap condition
WrapGet	LDX	#QTOP	;Successful dequeue
DeQDone	STX LDX DEC CLC RTS	GetPtr TempX QCount	Record new get pointer Restore X register Decrement Q counter Clear carry bit Return Unsuccessful dequeue
QEmpty	LDX SEC RTS	TempX	Restore X register Set carry bit Return
* Vector def			
*	ORG	RESETVEC	

FDB

Start

MACQ (Multiple Access Circular Queue)

A multiple access circular queue (or circular buffer) is a modified version of the queue data structure. It is a fixed-length, order-preserving data structure, and always contains the most recent entries. It is useful for data flow problems, when only the latest data is of interest. Once initialized it is always full, and a write operation always discards the oldest data.

Figure 4 depicts a MACQ.

Reading and Writing

After being initially filled, a write operation will place new data at the top of the MACQ, and shift existing data downward. The last byte will be discarded, so the result is the latest data existing in the buffer.

A read operation is non-destructive and can return any number of data bytes desired from the MACQ.




Figure 4. Result of a MACQ Write

Applications

A MACQ is useful for data streams which require the latest data and can afford to have a destructive write operation. For example, to predict the weather a forecaster might use temperature readings from the last five days to predict the next day's temperature. Daily temperature readings can be recorded in a MACQ, so the latest data is available.

MACQs are also useful for digital filters. For example, they can be used to calculate a second derivative, running average, etc.

Example

Semiconductor, Inc.

Listing 6. MACQ illustrates the implementation of a MACQ or circular buffer. This could be effectively used for storing A/D converter readings. In this way, the latest A/D conversion results would be accessible through the circular buffer.

Listing 6. MACQ

- *------
- * Illustrates an implementation of a multiple-access circular queue. (MACQ)
- * The MACQ is a fixed-length, order-preserving, indexable data structure.
- * Once initialized, the MACQ is always full.
- * A write to the MACQ is destructive, discarding the oldest data.
- * A read from the MACQ is non-destructive. For the 705P6A
- *-----
- * Register definitions
- * Memory map definitions

RAMSPACE	EQU	\$50
ROMSPACE	EQU	\$100
RESETVEC	EOU	\$1FFE

- * MACQueue data structure definitions
- * These three equates defines the data structure
- * To change the queue, change the data structure, and not the code.

QSIZE	EQU	8	;Maximum Q size
QTOP	EQU	\$A0	;Top of Q array
QBOT	EQU	QTOP+QSIZE-1	;Bottom of Q array

* RAM variables

	ORG	RAMSPACE	
TempA	RMB	1	
TempX	RMB	1	
TempData	RMB	1	
QPtr	RMB	1	;8-bit pointer

* Start of program code ROMSPACE Start EQU InitQ LDA #QBOT ;Initialize Q pointer STA QPtr * Write and read from the MACQ * A useful application of this would be to store A/D converter readings, so the latest * n readings are available. LDA #\$55 JSR WriteQ LDA #\$56 JSR WriteQ LDA #\$57 JSR WriteQ LDA #\$58 JSR WriteQ LDA #\$AA JSR WriteQ LDA #\$AB JSR WriteQ LDA #\$AC JSR WriteQ LDA #\$AD JSR WriteQ JSR WriteQ LDA JSR ReadQ LDA JSR ReadQ LDA #2 JSR ReadQ BRA Loop * Subroutines * WriteQ, A contains data to be written write is destructive on full Q, once initialized Q is always full. WriteQ STX ;Store X register value TempX

Freescale Semiconductor, Inc.

AN1752 — REV 1

LDX

CMPX

BEO

STA

QPtr

QFull

0,X

#QTOP-1

;Load Q pointer

;See if Q is full

;Store data to Q

		CX		;Decrement pointer
		STX	QPtr	;Store pointer
		BRA	WQDone	
* Once 1	MACQ is	; initialized	d, it's always f	full
QFull		STA	TempData	
		LDX	QBOT-1	Start shifting data down
SwapLoop		LDA	0,X	
		STA	1,X	
		DECX		
		CMPX	#QTOP	
		BHS	SwapLoop	
		LDX	#QTOP	
		LDA	TempData	
		STA	0,X	
WQDone		LDX	TempX	
		RTS		
*				
* ReadQ				
* A cont	tains q	queue index l	location to be 1	read returns value in A
*				
ReadQ	STX	TempX		
	ADD	#QTOP		;A = A (index) + QTOP pointer
	TAX			<pre>;X = address of desired value</pre>
	LDA	0,X		
	RTS			
*				
* Vector	defir	nitions		
*				
	ORG	RESETVEC		
	FDB	Start		

Tables

A table can be viewed as a vector of identically structured lists. A table is a common way of storing "lookup" data, such as display data or vector bytes.

Figure 5 shows an example of a table.




Figure 5. Table Representation

A table is commonly used to look up information. Table entries can be accessed with an offset from the base address of the table. Therefore, a read from a table is typically done by computing the offset of the desired data and accessing it using an indexed addressing mode.

Table Applications

The table data structure is common in MCU applications. One way of using tables is to perform character conversions. For example, a table can be used to convert binary numbers to BCD equivalents. For LCD (liquid crystal display) displays, an ASCII character byte may need to be converted to segment bitmaps for the display. A table could be used for this purpose.

Another application of a table is a "jump" table. This is a table of vector values which are addresses to be loaded and vectored to. Some program parameter can be converted to an offset into a jump table, so the appropriate vector is fetched for a certain input.

For example, in their memory maps Motorola MCUs have a built-in vector table, used for interrupt and exception processing. These vector tables allow preprogrammed addresses to be defined for certain MCU exceptions. When an exception occurs, a new program counter value is fetched from the appropriate table entry.

Another way of utilizing the table data structure is to store predefined values for lookup. An example of this is storing interpolation data in a table, to perform mathematical functions. This use of a table is documented in the application note, *M68HC08 Integer Math Routines*, Motorola document order number AN1219.

Another example involves using a table of sinusoidal values to produce sine wave output as in the application note *Arithmetic Waveform Synthesis with the HC05/08 MCUs*, Motorola document order number AN1222. If an equation to calculate data is CPU-intensive and can be approximated with discrete values, these values can be precalculated and stored in a table. In this way, a value can be quickly fetched, saving CPU time.

Table Example

Listing 7. Table is an example of the use of tables to convert ASCII data to LCD segment values.

Listing 7. Table

*_____

- * Code segment example of using a table to store LCD segment values
- * Could be used when 2 data registers define the segment values for a display position.
- * Takes in an ASCII character, converts it to an offset into the table of segment
- * values, and uses the offset to access the segment bitmap values.

*			
Loop	LDA	Character	;Load an ASCII character
	JSR	Convert	;Convert the character
	TAX		;Offset into table is in A
	LDA	0,X	;Load the first byte
	STA	LCD1	;Store to data register
	LDA	1,X	;Load the second byte
	STA	LCD2	;Store to data register
	BRA	Loop	;Repeat

- * Convert ASCII character byte in A to an offset value into the table of LCD segment
- * values. Valid ASCII values are (decimal): 32-47, 48-57, 65-90

*			
Convert	CMP	#!48	;Check for "special" character
	BLO	Special	
	CMP	#!65	;Check for numeric character
	BLO	Numeric	
Alpha	CMP	#!90	;Check for invalid value
	BHI	ConvError	

Freescale Semiconductor, Inc.

Application Note

	SUB	#!39	;Convert to table offset
	BRA	ConvDone	
Special	CMP	#!32	;Check for invalid value
	BLO	ConvError	
	SUB	#!32	Convert to table offset;
	BRA	ConvDone	
Numeric	CMP	#!57	Check for invalid value;
	BHI	ConvError	
	SUB	#!32	Convert to table offset;
	RA	ConvDone	
ConvError	CLRA		;Invalid value shows as blank
ConvDone	ROLA		;Multiply offset by 2
	RTS		;2 bytes data per LCD position
* Some chara * invalid, a	acters can n and will be	ot be displayed on displayed as a bla	ASCII character values 11-segment LCD, so they are marked as ink space.
Table	FDB	\$0000	;'''
	FDB	\$0000	;'!' INVALID
	FDB	\$0201	; ' " '
	FDB	\$0000	;'#' INVALID
	FDB	\$A5A5	;'\$'
	FDB	\$0000	;'%' INVALID
	FDB	\$0000	;'&' INVALID
	FDB	\$0001	;'''
	FDB	\$000A	;'('
	FDB	\$5000	;')'
	FDB	\$F00F	; ' * '
	FDB	\$A005	; ' + '
	FDB	\$0000	;',' INVALID
	FDB	\$2004	; ' = '
	FDB	\$0800	;'.'
	FDB	\$4002	;'/'
	FDB	\$47E2	;'0'
	FDB	\$0602	;'1'
	FDB	\$23C4	;'2'
	FDB	\$2784	;'3'
	FDB	\$2624	; ' 4 '
	FDB	\$21A8	; ′ 5 ′
	FDB	\$25E4	; '6'
	FDB	\$0700	; ′ 7 ′
	FDB	, \$27E4	; '8'
	FDB	\$27A4	; ′ 9 ′
	FDB	\$2764	; 'A'
	FDB	\$8785	;'B'
	FDB	, \$01E0	; 'C'
		40001	

FDB

FDB

FDB

\$8781

\$21E4

\$2164

;'D'

;'E' ;'F'

Application Note Linked Lists

FDB	\$05E4	;'G'
FDB	\$2664	;'H'
FDB	\$8181	;'I'
FDB	\$06C0	;'J'
FDB	\$206A	;'K'
FDB	\$00E0	;'L'
FDB	\$1662	;'M'
FDB	\$1668	;'N'
FDB	\$07E0	;'0'
FDB	\$2364	;'P'
FDB	\$07E8	; 'Q'
FDB	\$236C	;'R'
FDB	\$25A4	;'S'
FDB	\$8101	;'T'
FDB	\$06E0	;'U'
FDB	\$4062	; 'V'
FDB	\$4668	; 'W'
FDB	\$500A	; 'X'
FDB	\$9002	;'Y'
FDB	\$4182	;'Z'
EQU	*-Table	; End of table label

Linked Lists

EndTable

A list is a data structure whose elements may vary in precision. For example, a record containing a person's name, address, and phone number could be considered a list. A linked list is a group of lists, each of which contains a pointer to another list.

Figure 6 represents a linked list.




Figure 6. Linked List

Each list in the structure contains the same type of information, including a link to the next item in the list. The link might be an absolute address or an offset from some base address. In a doubly linked list, pointers are kept to both the next and the previous item in the list. A linked list can be traversed easily by simply following the pointers from one list to the next.

Linked List Applications A linked list is used traditionally to define a dynamically allocated database, in which the elements can be ordered or resorted by adjusting the links. However, in a small microcontroller, there are more appropriate applications of linked lists.

A linked list can be used as a structure for a command interpreter. Each command could contain the string of characters, an address of a subroutine to call on that command, and a link to the next command in the linked list. In this way, a command string could be input, searched for in a linked list, and appropriate action taken when the string is found.

State Machines

Another useful application of a linked list is to define a state machine. A state machine can be represented by a discrete number of states, each of which has an output and pointers to the next state(s). See **Figure 7**.

Application Note Linked Lists




Figure 7. State Machine

A state machine can be considered a Mealy or a Moore machine. A Mealy machine's output is a function of both its inputs and its current state. A Moore machine has an output dependent only on its current state.

This state machine model can be useful for controlling sequential devices such as vending machines, stepper motors, or robotics. These machines have a current internal state, receive input, produce output, and advance to the next state.

One can first model a process as a sequential machine, then convert this behavior to a linked-list structure and write an interpreter for it. An important goal is to be able to make modifications to the state machine by changing the data structure (linked list) and not the code.

State Machine Example

As an example, consider a traffic light controller which determines the light patterns for an intersection. Two light patterns are needed, one for the north/south directions and one for the east/west directions. Consider that the bulk of traffic travels on the north/south road, but sensors are placed at the east/west road intersection to determine when traffic needs to cross. See **Figure 8**.




Figure 8. Traffic Light Controller Example

This example can be modeled as a Moore state machine, with its output a function of its current state. The next state is a function of the current state and the state of the input. **Figure 9** shows a state graph for this example. The initial state will be a green light in the north/south direction and a red light in the east/west direction. The controller remains in this state, until input is seen in the east/west direction. The flow continues as shown in the diagram. The output shown in the diagram is a pattern for the light array to activate the lights for the state.

Application Note Linked Lists




Figure 9. Traffic-Light Controller State Graph

Simulation

This example can be simulated using LEDs and a 68HC705P6A MCU. A pushbutton switch can be used to simulate the input sensor. **Figure 10** illustrates the simulation circuit. Using six bits of an output port, a pattern can be generated to display the appropriate north/south and east/west lights (LEDs). **Table 1** shows the bitmap in this application.




Figure 10. Circuit Simulation of Traffic-Light Controller

Table 1. Traffic Light Bitmap for Port A

Bit Position	7	6	5	4	3	2	1	0
Meaning	Not	used	used North/South Signal		East/West Signal			
Wearing	Х	Х	G	Y	R	G	Υ	R

With the hardware in place, all that is left is to define the state machine in software. This can be done by implementing a linked-list data structure and the code to access and interpret the machine.

For this particular example, each list in the data structure defines the current state of the traffic light. Each list contains:

- The byte which is the bitmap for the lights.
- 2. A delay value; the time the controller remains in the state
- 3. The next state pointer for an input of 0
- The next state pointer for an input of 1

The main loop of the program should execute the program flow charted in **Figure 11**. The software for this simulated traffic light controller is documented in **Listing 8**. **Traffic Controller State Machine**.




Figure 11. State Machine Program Flow

Listing 8. Traffic Controller State Machine

*-----

* Traffic light controller example Illustrates a linked list implementation of a state

* machine For the 705P6A

* Register definitions

PORTA EQU \$00

PORTD EQU \$03

DDRA EQU \$04

DDRD EQU \$07

* Memory map definitions
RAMSPACE EQU \$50
ROMSPACE EQU \$100
RESETVEC EQU \$1FFE
* RAM variables

RAM Variables

ORG RAMSPACE

Freescale Semiconductor, Inc.

Application Note

ГеmpA	RMB	1	
ΓempX	RMB	1	
	program code 		
		ROMSPACE	
Start	LDA	#\$00	
	STA	PORTA	;Predefine output levels
	LDA	#\$FF	<u>-</u>
		DDRA	;Make Port A all outputs
	BCLR	7,PORTD	;Make Port D pin 0 an input
		#INITST	;Index initial state
Loop			Get light pattern
		PORTA	Output light pattern
		STATES+DELAY,X	
		SecDelay	;Cause delay
		7,PORTD,In0	
In1	LDX	STATES+NEXT0,X	Get next state offset
		Loop	;(input = 1)
In0	LDX	STATES+NEXT1,X	Get next state offset
	DD 3	Loop	;(input = 0)
* DATA STRU * Offsets a	UCTURE FOR STATE Mand base address so	ACHINE LINKED LIS cheme is adequate	T (05) for a small table (<255 bytes)
* DATA STRU * Offsets a	UCTURE FOR STATE Mind base address so	ACHINE LINKED LIS	T (05) for a small table (<255 bytes)
* DATA STRU * Offsets a *	UCTURE FOR STATE Mand base address so	ACHINE LINKED LIS cheme is adequate 	T (05) for a small table (<255 bytes)
* DATA STRU * Offsets a * LIGHTS DELAY	UCTURE FOR STATE MA and base address so EQU EQU	ACHINE LINKED LIS cheme is adequate 0 0	T (05) for a small table (<255 bytes)
* DATA STRU * Offsets a * LIGHTS DELAY NEXTO	UCTURE FOR STATE MA and base address so EQU EQU EQU	ACHINE LINKED LIS cheme is adequate 0 1	T (05) for a small table (<255 bytes) ;Offset for light pattern ;Offset for time delay ;Offset for pointer 0
* DATA STRU * Offsets a * LIGHTS DELAY NEXTO NEXT1	UCTURE FOR STATE MA and base address so EQU EQU EQU EQU EQU	ACHINE LINKED LIS cheme is adequate 0 1 2	T (05) for a small table (<255 bytes) ;Offset for light pattern ;Offset for time delay ;Offset for pointer 0 ;Offset for pointer 1
* DATA STRU * Offsets a * LIGHTS DELAY NEXTO NEXT1 STATES	UCTURE FOR STATE MA and base address so EQU EQU EQU EQU EQU EQU EQU	ACHINE LINKED LIS cheme is adequate 0 1 2 3	T (05) for a small table (<255 bytes) ;Offset for light pattern ;Offset for time delay ;Offset for pointer 0 ;Offset for pointer 1 ;Base address of states
* DATA STRU * Offsets a * LIGHTS DELAY NEXTO NEXT1 STATES INITST	UCTURE FOR STATE MA and base address so EQU EQU EQU EQU EQU EQU EQU EQU	ACHINE LINKED LIS cheme is adequate 0 1 2 3 * *-STATES	T (05) for a small table (<255 bytes) ;Offset for light pattern ;Offset for time delay ;Offset for pointer 0 ;Offset for pointer 1 ;Base address of states ;Initial state offset
* DATA STRU * Offsets a * LIGHTS DELAY NEXTO NEXT1 STATES INITST * North/Sou	EQU	ACHINE LINKED LIS cheme is adequate 0 1 2 3 * *-STATES ast/West red ligh	T (05) for a small table (<255 bytes) ;Offset for light pattern ;Offset for time delay ;Offset for pointer 0 ;Offset for pointer 1 ;Base address of states ;Initial state offset
* DATA STRU * Offsets a * LIGHTS DELAY NEXTO NEXT1 STATES INITST	EQU	ACHINE LINKED LIS cheme is adequate 0 1 2 3 * *-STATES ast/West red ligh *-STATES	T (05) for a small table (<255 bytes) ;Offset for light pattern ;Offset for time delay ;Offset for pointer 0 ;Offset for pointer 1 ;Base address of states ;Initial state offset t ;Offset into STATES
* DATA STRU * Offsets a * LIGHTS DELAY NEXTO NEXT1 STATES INITST * North/Sou	EQU	ACHINE LINKED LIS cheme is adequate 0 1 2 3 * *-STATES ast/West red ligh *-STATES %11011110	T (05) for a small table (<255 bytes) ;Offset for light pattern ;Offset for time delay ;Offset for pointer 0 ;Offset for pointer 1 ;Base address of states ;Initial state offset t ;Offset into STATES ;Output for state
* DATA STRU * Offsets a * LIGHTS DELAY NEXTO NEXT1 STATES INITST * North/Sou	EQU	ACHINE LINKED LIS cheme is adequate 0 1 2 3 * *-STATES ast/West red ligh *-STATES \$11011110	T (05) for a small table (<255 bytes) ;Offset for light pattern ;Offset for time delay ;Offset for pointer 0 ;Offset for pointer 1 ;Base address of states ;Initial state offset t ;Offset into STATES ;Output for state ;Delay for state
* DATA STRU * Offsets a * LIGHTS DELAY NEXTO NEXT1 STATES INITST * North/Sou	EQU	ACHINE LINKED LIS cheme is adequate 0 1 2 3 * *-STATES ast/West red ligh *-STATES \$11011110 !10 NSG	T (05) for a small table (<255 bytes) ;Offset for light pattern ;Offset for time delay ;Offset for pointer 0 ;Offset for pointer 1 ;Base address of states ;Initial state offset t ;Offset into STATES ;Output for state ;Delay for state ;Next state for input of 0
* DATA STRU * Offsets a * LIGHTS DELAY NEXTO NEXT1 STATES INITST * North/Sou	UCTURE FOR STATE MA and base address so EQU EQU EQU EQU EQU EQU EQU EQU TCB FCB FCB FCB	ACHINE LINKED LIS cheme is adequate 0 1 2 3 * *-STATES ast/West red ligh *-STATES %11011110 !10 NSG NSY	T (05) for a small table (<255 bytes) ;Offset for light pattern ;Offset for time delay ;Offset for pointer 0 ;Offset for pointer 1 ;Base address of states ;Initial state offset t ;Offset into STATES ;Output for state ;Delay for state
* DATA STRU * Offsets a * LIGHTS DELAY NEXTO NEXT1 STATES INITST * North/Sou	EQU	ACHINE LINKED LIS cheme is adequate 0 1 2 3 * *-STATES ast/West red ligh *-STATES %11011110 !10 NSG NSY light	T (05) for a small table (<255 bytes) ;Offset for light pattern ;Offset for time delay ;Offset for pointer 0 ;Offset for pointer 1 ;Base address of states ;Initial state offset t ;Offset into STATES ;Output for state ;Delay for state ;Next state for input of 0
* DATA STRU * Offsets a * LIGHTS DELAY NEXTO NEXT1 STATES INITST * North/Sou	EQU	ACHINE LINKED LIS cheme is adequate 0 1 2 3 * *-STATES ast/West red ligh *-STATES %11011110 !10 NSG NSY light *-STATES	T (05) for a small table (<255 bytes) ;Offset for light pattern ;Offset for time delay ;Offset for pointer 0 ;Offset for pointer 1 ;Base address of states ;Initial state offset t ;Offset into STATES ;Output for state ;Delay for state ;Next state for input of 0
* DATA STRU * Offsets a * LIGHTS DELAY NEXTO NEXT1 STATES INITST * North/Sou	EQU	ACHINE LINKED LIS cheme is adequate 0 1 2 3 * *-STATES ast/West red ligh *-STATES \$11011110 !10 NSG NSY light *-STATES \$11101110	T (05) for a small table (<255 bytes) ;Offset for light pattern ;Offset for time delay ;Offset for pointer 0 ;Offset for pointer 1 ;Base address of states ;Initial state offset t ;Offset into STATES ;Output for state ;Delay for state ;Next state for input of 0
* DATA STRU * Offsets a * LIGHTS DELAY NEXTO NEXT1 STATES INITST * North/Sou	EQU	ACHINE LINKED LIS cheme is adequate 0 1 2 3 * *-STATES ast/West red ligh *-STATES %11011110 !10 NSG NSY light *-STATES %11101110 !5	T (05) for a small table (<255 bytes) ;Offset for light pattern ;Offset for time delay ;Offset for pointer 0 ;Offset for pointer 1 ;Base address of states ;Initial state offset t ;Offset into STATES ;Output for state ;Delay for state ;Next state for input of 0
* DATA STRU * Offsets a * LIGHTS DELAY NEXTO NEXT1 STATES INITST * North/Sou	EQU	ACHINE LINKED LIS cheme is adequate 0 1 2 3 * *-STATES ast/West red ligh *-STATES %11011110 !10 NSG NSY light *-STATES %11101110 !5 NSR	T (05) for a small table (<255 bytes) ;Offset for light pattern ;Offset for time delay ;Offset for pointer 0 ;Offset for pointer 1 ;Base address of states ;Initial state offset t ;Offset into STATES ;Output for state ;Delay for state ;Next state for input of 0
* DATA STRU * Offsets a * LIGHTS DELAY NEXTO NEXT1 STATES INITST * North/Sou	EQU	ACHINE LINKED LIS cheme is adequate 0 1 2 3 * *-STATES ast/West red ligh *-STATES %11011110 !10 NSG NSY light *-STATES %1101110 !5 NSR NSR	T (05) for a small table (<255 bytes) ;Offset for light pattern ;Offset for time delay ;Offset for pointer 0 ;Offset for pointer 1 ;Base address of states ;Initial state offset t ;Offset into STATES ;Output for state ;Delay for state ;Next state for input of 0
* DATA STRU * Offsets a * LIGHTS DELAY NEXTO NEXT1 STATES INITST * North/Sou NSG * N/S yello NSY	EQU	ACHINE LINKED LIS cheme is adequate 0 1 2 3 * *-STATES ast/West red ligh *-STATES %11011110 !10 NSG NSY light *-STATES %11101110 !5 NSR NSR NSR	T (05) for a small table (<255 bytes) ;Offset for light pattern ;Offset for time delay ;Offset for pointer 0 ;Offset for pointer 1 ;Base address of states ;Initial state offset t ;Offset into STATES ;Output for state ;Delay for state ;Next state for input of 0
* DATA STRU * Offsets a * LIGHTS DELAY NEXTO NEXT1 STATES INITST * North/Sou	EQU	ACHINE LINKED LIS cheme is adequate 0 1 2 3 * *-STATES ast/West red ligh *-STATES %11011110 !10 NSG NSY light *-STATES %11101110 !5 NSR NSR NSR ight *-STATES	T (05) for a small table (<255 bytes) ;Offset for light pattern ;Offset for time delay ;Offset for pointer 0 ;Offset for pointer 1 ;Base address of states ;Initial state offset t ;Offset into STATES ;Output for state ;Delay for state ;Next state for input of 0
* DATA STRU * Offsets a * LIGHTS DELAY NEXTO NEXT1 STATES INITST * North/Sou NSG * N/S yello NSY	EQU	ACHINE LINKED LIS cheme is adequate 0 1 2 3 * *-STATES ast/West red ligh *-STATES %11011110 !10 NSG NSY light *-STATES %11101110 !5 NSR NSR NSR	T (05) for a small table (<255 bytes) ;Offset for light pattern ;Offset for time delay ;Offset for pointer 0 ;Offset for pointer 1 ;Base address of states ;Initial state offset t ;Offset into STATES ;Output for state ;Delay for state ;Next state for input of 0

* E/W yellow l	FCB ight, N/S red EQU FCB FCB FCB FCB	EWY light *-STATES %11110101 !5 ;Delay for state NSG NSG
* Delay subrou	tines	
*		
* Cause a dela	y of ~(1 seco	nd * Accumulator value) @ fop = 1MHz
SecDelay	CMP	#\$00
	BEQ	SecDone
	JSR	Delay0
	JSR	Delay0
	DECA	
	BRA	SecDelay
SecDone	RTS	
*		
* Cause a dela	y of ~1/2 of	a second
Delay0	STX	TempX
-	LDX	#\$B2
DLoop0	CMPX	#\$00
	BEQ	DDone0
	JSR	Delay1
	DECX	
	BRA	DLoop0
DDone0	LDX	TempX
	RTS	
*		
* Cause about	2.8msec delay	@ fop of 1MHz
Delay1	STA	TempA
	LDA	#\$FF
DLoop1	CMP	#\$00
	BEQ	DDone1
	DECA	
	BRA	DLoop1
DDone1	LDA	TempA
	RTS	
** * Vector defin	itions	
*		
		ETVEC
	FDB Sta	rt

Freescale Semiconductor, Inc.

Application Note

Conclusion

The use of data structures is not necessarily limited to large, complicated computers. Although the data structure is a powerful concept in such a context, the same principles can be applied to smaller processors such as 8-bit microcontrollers.

The code to implement these data structures does not necessarily have to be complex or confusing. The goal of programming should be to modularize commonly used functions, so that they may be reused in other applications with a minimal amount of modification.

The appropriate use of data structure concepts can improve the static and dynamic performance of an MCU application, without affecting its portability or legibility.

Motorola reserves the right to make changes without further notice to any products herein. Motorola makes no warranty, representation or guarantee regarding the suitability of its products for any particular purpose, nor does Motorola assume any liability arising out of the application or use of any product or circuit, and specifically disclaims any and all liability, including without limitation consequential or incidental damages. "Typical" parameters which may be provided in Motorola data sheets and/or specifications can and do vary in different applications and actual performance may vary over time. All operating parameters, including "Typicals" must be validated for each customer application by customer's technical experts. Motorola does not convey any license under its patent rights nor the rights of others. Motorola products are not designed, intended, or authorized for use as components in systems intended for surgical implant into the body, or other applications intended to support or sustain life, or for any other application in which the failure of the Motorola product could create a situation where personal injury or death may occur. Should Buyer purchase or use Motorola products for any such unintended or unauthorized application, Buyer shall indemnify and hold Motorola and its officers, employees, subsidiaries, affiliates, and distributors harmless against all claims, costs, damages, and expenses, and reasonable attorney fees arising out of, directly or indirectly, any claim of personal injury or death associated with such unintended or unauthorized use, even if such claim alleges that Motorola was negligent regarding the design or manufacture of the part. Motorola and

How to reach us:

USA/EUROPE/Locations Not Listed: Motorola Literature Distribution; P.O. Box 5405, Denver, Colorado 80217. 1-303-675-2140 or 1-800-441-2447

JAPAN: Motorola Japan Ltd.; SPS, Technical Information Center, 3-20-1, Minami-Azabu, Minato-ku, Tokyo 106-8573 Japan. 81-3-3440-3569

ASIA/PACIFIC: Motorola Semiconductors H.K. Ltd.; Silicon Harbour Centre, 2 Dai King Street, Tai Po Industrial Estate, Tai Po, N.T., Hong Kong. 852-26668334

Technical Information Center: 1-800-521-6274

HOME PAGE: http://www.motorola.com/semiconductors/

© Motorola, Inc., 1998, 2001

